
Lecture 11
Linear systems of equations I

1 Introduction
In this lecture we consider ways to solve a linear system  Ax=b  for  x when given  A and  b.
Writing out the components our system has the form

(
a11 a12 a13 ⋯ a1n

a21 a22 a23 ⋯ a2 n

a31 a32 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋱ ⋮
an1 an2 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

) (1)

Each row of A corresponds to a single linear equation in the unknown x values. The ith equation
is

a i 1 x1+a i 2 x2+⋯+a in xn=b i

and the  jth equation is

a j 1 x1+a j 2 x2+⋯+a j n xn=b j

We will use the following facts to transform the system (1) into a form in which the solution is
trivially apparent or at least can be easily calculated.

Fact 1:  We can scale any equation by a non-zero constant  ( c≠0 ) without  changing the
solution

a i 1 x1+a i 2 x2+⋯+a in xn=b i  ←  ca i 1 x1+ca i 2 x2+⋯+ca in xn=cbi

Fact 2: We can replace an equation by its sum or difference with another equation without
changing the solution

a i 1 x1+a i 2 x2+⋯+a in xn=b i  ←  (a i 1+a j 1) x1+(ai 2+a j 2) x2+⋯+(ai n+a jn)xn=bi+b j

2 Gauss-Jordan elimination
If A was the identity matrix

(
1 0 0 ⋯ 0
0 1 0 ⋯ 0
0 0 1 ⋯ 0
⋮ ⋮ ⋮ ⋮ ⋮
0 0 0 ⋯ 1

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

) (2)

the solution would be trivial

xk=bk

Gauss-Jordan elimination is a process to convert an arbitrary system (1) into the trivial system
(2). Since we want to end up with a11=1 , we use Fact 1 to multiply the first row of A and b by
1/a11

a1 j ←a1 j / a11 , j=1,2,… , n  , b1 ←b1/a11

to obtain the form 

(
1 a12 a13 ⋯ a1n

a21 a22 a23 ⋯ a2n

a31 a32 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋮ ⋮
an1 an 2 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
(note  a12 , a13  etc.  will  have changed values).  Now we use Fact 2 to eliminate  the elements
a21 , a31 ,… , an 1  in the first column by subtracting a i 1  times the first row from the ith row 

a ij ←a ij−a i 1 a1 j , j=1,2,… , n  , bi ←bi−ai 1 b1

for i=2,3,… , n  resulting in

(
1 a12 a13 ⋯ a1n

0 a22 a23 ⋯ a2n

0 a32 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋮ ⋮
0 an2 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
The first column is now in the desired form. Let's move to the second column. Since we want to
end up with a22=1 , use Fact 1 to multiply the second row by 1/a22

a2 j ←a2 j /a22 , j=2,3,… , n  , b2 ←b2/a22

Note that we don't bother with j=1  since a21=0 . Then we use Fact 2 to eliminate all elements
a i 2 , i≠2

a ij ←a ij−a i 2 a2 j , j=2,3,… , n  , bi ←bi−ai 2b2

Again we don't bother with j=1  since a21=0 . We end up with

(
1 0 a13 ⋯ a1n

0 1 a23 ⋯ a2n

0 0 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋮ ⋮
0 0 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
We now move on to the third column and so on, continuing until we have the form shown in (2). 

Notice that the element bi  is transformed in the same manner as the elements a ij . This suggests
that we form an augmented matrix as A plus b as an extra column and then simply transform the
a values as a whole. Let's call the augmented matrix 

~
A . Then

~A=(
a11 a12 a13 ⋯ a1 n b1

a21 a22 a23 ⋯ a2 n b2

a31 a32 a33 ⋯ a3 n b3

⋮ ⋮ ⋮ ⋱ ⋮ ⋮
an 1 an 2 an 3 ⋯ ann bn

)=(
a11 a12 a13 ⋯ a1 n a1 n+1

a21 a22 a23 ⋯ a2 n a2 n+1

a31 a32 a33 ⋯ a3 n a3 n+1

⋮ ⋮ ⋮ ⋱ ⋮ ⋮
an 1 an 2 an 3 ⋯ ann an n+1

)
We simply replace a ij ←a ij−ai 2 a2 j , j=2,3,… , n  by a ij ←a ij−a i 2 a2 j , j=2,3,… , n+1  and the b
values are automatically taken care of.  Our algorithm can now be stated as

Gauss-Jordan elimination algorithm

  Form the augmented matrix ~A=(A ,b)

  for k=1,2,… , n

      for j=n+1,n , n−1,… , k

          akj ←akj /akk

      for i=1,2,… , n

          if i≠k  (we don't want a row to “eliminate” itself)

              for j=n+1,n , n−1,… , k

                  a ij ←a ij−aik akj

When the loops are complete, the last column of 
~
A  will be the solution vector x. 

Example 1: Consider the system Ax=b with

A=(1 2
3 4)  , b=( 4

10)
We form the augmented matrix

~A=(1 2 4
3 4 10)

The first pivot, a11 , is already 1. We now subtract 3 times row 1 from row 2

~A=(1 2 4
0 −2 −2)

The second pivot is a22=−2 . We divide row 2 by this to get

~A=(1 2 4
0 1 1)

We now subtract 2 times row 2 from row 1 to obtain

~A=(1 0 2
0 1 1)

The last column is the solution x.

The function linearGaussJordan shown in the Appendix is a Scilab implementation of this
algorithm.

Now, suppose we solve the following problems one at a time using this method

A x1=(
1
0
0
⋮
0
)  , A x2=(

0
1
0
⋮
0
)  , A x3=(

0
0
1
⋮
0
)  , …  , A xn=(

0
0
0
⋮
1
)

Then

A (x1 ,x2 ,x3 ,… ,xn)=(
1 0 0 ⋯ 0
0 1 0 ⋯ 0
0 0 1 ⋯ 0
⋮ ⋮ ⋮ ⋱ 0
0 0 0 ⋯ 1

)
or  AX=I  where  X is the matrix with column vectors  x1 ,x2 ,… ,xn . It follows that  X is the
inverse of  A. We don't have to solve  n problems one at a time, however. The operations we
perform with the a ij  values will be the same regardless of the right-hand vector b. Therefore, we
can simply form the augmented matrix

~A=(
a11 a12 a13 ⋯ a1n 1 0 0 ⋯ 0
a21 a22 a23 ⋯ a2n 0 1 0 ⋯ 0
a31 a32 a33 ⋯ a3n 0 0 1 ⋯ 0
⋮ ⋮ ⋮ ⋱ ⋮ ⋮ ⋮ ⋮ ⋱ ⋮

an1 an 2 an3 ⋯ ann 0 0 0 ⋯ 1
)

and perform Gauss-Jordan elimination on 
~
A  to effectively solve these n problems “in parallel.”

When complete, the last  n columns of  ~A  will be the inverse A−1 . In the Appendix, program
linearGaussJordanInverse gives a Scilab implementation of this algorithm. Note the
slight differences between this and linearGaussJordan.

Notice that the Gauss-Jordan elimination algorithm consists of three nested levels of for loops.
Each (runs roughly speaking) over on order of  n values.  It follows that  the total  number of
operations is on the order of n⋅n⋅n=n3 . This gives a measure of the computational complexity
and therefore of the amount of cpu time required for the algorithm. For this reason you will often
see  statements  of  the  form  “matrix  inversion  is  an  O(n3

)  operation,  where  the  “big  O”
represents “order of.” 

3 Gaussian elimination
Gauss-Jordan elimination is a logical way to solve Ax=b  or to find A−1 . However, there are
faster and more robust methods.  In particular  we can solve  Ax=b  with only about 1/3 the
number of operations using the so-called LU decomposition that we will  develop in the next
lecture. The price we pay is that getting the solution is more convoluted than it is for Gauss-

Jordan elimination; it involves various “substitution” operations. Here we'll introduce this idea
by considering the so-called Gaussian elimination algorithm. 

In Gauss-Jordan elimination we “zero-out” all elements of A except those on the diagonal, and
we normalize  the  diagonal  elements  to  1.  In Gaussian elimination  we don't  bother  with  the
elements above the diagonal; we only zero-out the elements below the diagonal. We also don't
bother to normalize the diagonal elements to 1. The result is a system in the form

(
a11 a12 a13 ⋯ a1n

0 a22 a23 ⋯ a2 n

0 0 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋱ ⋮
0 0 0 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

) (3)

The matrix is in upper-triangular form; all elements below the diagonal are zero. The algorithm
to achieve this is an obvious variation of Gauss-Jordan elimination.

Gaussian elimination algorithm

  Form the augmented matrix 
~
A=(A , b)

  for k=1,2,… , n−1

      for i=k+1, k+2,… , n

          for j=n+1, n , n−1,… , k

              a ij ←a ij−akj a ik /akk

Since it has three nested for loops, each running over on the order of  n values, this is also an
O(n3

)  process. A problem with (3) is that the solution is not obvious (as it is for (2)) except for
the last row which gives us

ann xn=bn → xn=bn/ann

So, xn  is easy to get. The next-to-last equation is

an−1 , n−1 xn−1+an−1 ,n xn=bn−1

But, we already know xn , so we can solve for

xn−1=
1

an−1 , n−1
(bn−1−an−1 , n xn)

The second-to-last equation is

an−2 ,n−2 xn−2+an−2 ,n−1 xn−1+an−2 , n xn=bn−2

and since we've already solved for xn−1 , xn  we can solve for the single remaining unknown

xn−2=
1

an−2 , n−2
(bn−2−[an−2 ,n−1 xn−1+an−2 , n xn ])

Continuing to work our way up row-by-row we have the following algorithm.

EE 221 Numerical Computing Scott Hudson 2015-08-18



Lecture 11: Linear systems of equations I 6/13

Back-substitution algorithm

  If A is upper-triangular, the solution to Ax=b  is given by

  for i=n , n−1, n−2,… , 1

      x i=
1
a ii

(bi− ∑
j=i+1

n

a ij x j)  (summation is zero when i=n )

A Scilab implementation of Gaussian elimination followed by back-substitution appears in the
Appendix  as  linearGaussian.  This  algorithm has  two nested  for  loops  (i  and  j)  and is
therefore an O(n2

)  process. This tells us that, at least for large matrices (large n values), back-
substitution is much faster than Gaussian elimination or Gauss-Jordan elimination. The fact that
we've had to add this step, therefore, is not of much concern regarding cpu time. 

4 The need for “pivoting”
Looking over the Gauss-Jordan and Gaussian elimination algorithms it is clear that the only way
they can “go wrong” is if akk=0  for some value of k since we would then be trying to divide by
zero. Note that when we get to this point in either algorithm, in general  akk  will not have the
same value it had in the original matrix A, but will have been modified by various subtractions
performed in the algorithm. Therefore, even if none of the diagonal elements of A are zero, we
can still end up with akk=0  at some stage in the algorithm. Furthermore, even if akk  is non-zero
but very small this will still cause problems by amplifying accumulated round-off error by a large
factor of 1/akk . 

The diagonal elements akk  are called pivots, and the way to avoid the problem of a small or zero
pivot  is  through  pivoting –  swapping  two  rows  or  two  columns  of  the  augmented  matrix.
Pivoting is a critical requirements for a robust algorithm, and without it the code given in the
Appendix is not generally reliable.

Let's again consider the general problem Ax=b  written out in component form

(
a11 a12 a13 ⋯ a1n

a21 a22 a23 ⋯ a2 n

a31 a32 a33 ⋯ a3n

⋮ ⋮ ⋮ ⋱ ⋮
an1 an2 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
Each row represents a single linear equation in the unknowns. In what way will the solution
change if we swap, say, the first and third rows of both A and b?

(
a31 a32 a33 ⋯ a3n

a21 a22 a23 ⋯ a2 n

a11 a12 a13 ⋯ a1n

⋮ ⋮ ⋮ ⋱ ⋮
an1 an2 an3 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b3

b2

b1

⋮
bn

)
The answer is that it won't. We've just rearranged the same n equations in n unknowns. It doesn't

matter in what order we write them; the solution will remain the same. 

Fact 3: Any two rows of the augmented matrix  ~
A  can be swapped without changing the

solution vector x. 

What about swapping columns? Say we swap the first and third columns of A.

(
a13 a12 a11 ⋯ a1n

a23 a22 a21 ⋯ a2 n

a33 a32 a31 ⋯ a3n

⋮ ⋮ ⋮ ⋱ ⋮
an3 an2 an1 ⋯ ann

)(
x1

x2

x3

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
Thinking of Ax as a linear combination of the columns of A, the ith column of A gets multiplied
by x i . Swapping the columns is equivalent to relabeling the two corresponding components of x.
In other words if we write the system as

(
a13 a12 a11 ⋯ a1n

a23 a22 a21 ⋯ a2 n

a33 a32 a31 ⋯ a3n

⋮ ⋮ ⋮ ⋱ ⋮
an3 an2 an1 ⋯ ann

)(
x3

x2

x1

⋮
xn

)=(
b1

b2

b3

⋮
bn

)
it is just a rearrangement of our original system. Provided we remember to swap x3 , x1  at the
end this system will give us the same solution as the original system. 

Fact 4: Any two columns of the matrix  A  can be swapped without changing the solution
vector x, except for a reordering of its elements.

The idea of “full pivoting” is that when we come to the place in our algorithm where we will be
dividing by akk , we examine all the ways we can swap rows and columns so as to replace akk

by the largest magnitude value possible. Provided we do the bookkeeping correctly, this won't
change our solution but it will make it numerically robust. 

The difficultly with full pivoting is that for each loop there are a lot of possible row-plus-column
swaps, and finding the best involves many absolute-value tests in each loop. Moreover, if we
swap columns we need to keep track of this  in  order to “untangle” the  x values at  the end.
However, if we limit ourselves to a single row swap, which is called “partial pivoting” or “row
pivoting,” then we need only check the magnitudes of the elements in the  kth column, and we
don't have to keep track of the swaps. Experience shows that partial  pivoting is sufficient to
produce reliable algorithms. Almost always in practice, therefore, only partial pivoting is used.
From here on when we talk about “pivoting” we mean “partial pivoting”  – swapping rows only.
Adding partial  pivoting  to  the  linearGaussian code  results  in  the  function
linearGaussianPivot given in the Appendix. This is a serviceable routine. 

5 References
1. Golub, G.H. and C.F. Van Loan, Matrix Computations, Johns Hopkins University Press,

1983, ISBN: 0-8018-3011-7.

6 Appendix – Scilab code

6.1 Gauss-Jordan elimination
0001  //////////////////////////////////////////////////////////////////////
0002  // linearGaussJordan.sci
0003  // 2014-06-23, Scott Hudson, for pedagogic purposes
0004  // Solves Ax=b for x using Gauss-Jordan elimination.
0005  // No pivoting is performed.
0006  //////////////////////////////////////////////////////////////////////
0007  function x=linearGaussJordan(A, b)
0008    n = length(b);
0009    A = [A,b]; //form augmented matrix
0010    for k=1:n //A(k,k) is the pivot
0011      for j=n+1:-1:k //normalize row k so A(k,k)=1
0012        A(k,j) = A(k,j)/A(k,k); 
0013      end
0014      for i=1:n //eliminate a(i,k) for all i~=k
0015        if (i~=k) //a Pivot does not eliminate itself
0016          for j=n+1:-1:k
0017            A(i,j) = A(i,j)-A(k,j)*A(i,k);
0018          end
0019        end
0020      end //i loop
0021    end //k loop
0022    x = A(:,n+1); //last column of augmented matrix is now x
0023  endfunction

6.2 Matrix inverse using Gauss-Jordan elimination
0001  //////////////////////////////////////////////////////////////////////
0002  // linearGaussJordanInverse.sci
0003  // 2014-06-23, Scott Hudson, for pedagogic purposes
0004  // Forms inverse of matrix A using Gauss-Jordan elimination.
0005  // No pivoting is performed.
0006  //////////////////////////////////////////////////////////////////////
0007  function Ainv=linearGaussJordanInverse(A)
0008    n = size(A,'r');
0009    A = [A,eye(A)]; //form augmented matrix
0010    for k=1:n //A(k,k) is the pivot
0011      for j=2*n:-1:k //normalize row k so A(k,k)=1
0012        A(k,j) = A(k,j)/A(k,k); 
0013      end
0014      for i=1:n //eliminate a(i,k) for all i~=k
0015        if (i~=k) //a Pivot does not eliminate itself
0016          for j=2*n:-1:k
0017            A(i,j) = A(i,j)-A(k,j)*A(i,k);
0018          end
0019        end
0020      end //i loop
0021    end //k loop
0022    Ainv = A(:,n+1:2*n); //last column of augmented matrix is now x
0023  endfunction

6.3 Gaussian elimination
0001  //////////////////////////////////////////////////////////////////////
0002  // linearGaussian.sci
0003  // 2014-06-25, Scott Hudson, for pedagogic purposes
0004  // Solves Ax=b for x using Gaussian elimination and backsubstitution.
0005  // No pivoting is performed.
0006  //////////////////////////////////////////////////////////////////////
0007  function x=linearGaussian(A, b)
0008    n = length(b);
0009    A = [A,b]; //form augmented matrix
0010    //Gaussian elimination loop
0011    for k=1:n-1 //A(k,k) is the pivot
0012      for i=k+1:n //eliminate a(i,k) for all i>k
0013        for j=n+1:-1:k
0014          A(i,j) = A(i,j)-A(k,j)*A(i,k)/A(k,k);
0015        end
0016      end //i loop
0017    end //k loop
0018    //Backsubstitution loop
0019    x = zeros(n,1);
0020    x(n) = A(n,n+1)/A(n,n); 
0021    for i=n-1:-1:1
0022      x(i) = A(i,n+1);
0023      for j=i+1:n
0024        x(i) = x(i)-A(i,j)*x(j);
0025      end
0026      x(i) = x(i)/A(i,i);
0027    end
0028  endfunction

6.4 Gaussian elimination with partial pivoting
0001  //////////////////////////////////////////////////////////////////////
0002  // linearGaussianPivot.sci
0003  // 2014-06-25, Scott Hudson, for pedagogic purposes
0004  // Solves Ax=b for x using Gaussian elimination and backsubstitution.
0005  // Partial pivoting is performed.
0006  //////////////////////////////////////////////////////////////////////
0007  function x=linearGaussianPivot(A, b)
0008    n = length(b);
0009    A = [A,b]; //form augmented matrix
0010    //Gaussian elimination loop
0011    for k=1:n-1 //A(k,k) is the pivot
0012      //see if there is a larger pivot below this in the kth column
0013      Amax = abs(A(k,k));
0014      imax = k;
0015      for i=k+1:n
0016        if abs(A(i,k))>Amax
0017          Amax = abs(A(i,k));
0018          imax = i;
0019        end
0020      end
0021      if (imax~=k) //we found a larger pivot, swap rows
0022        w = A(k,:); //copy the kth row
0023        A(k,:) = A(imax,:); //replace it with the imax row
0024        A(imax,:) = w; //replace the imax row with the original kth row
0025      end
0026      //pivoting complete
0027      for i=k+1:n //eliminate a(i,k) for all i>k
0028        for j=n+1:-1:k
0029          A(i,j) = A(i,j)-A(k,j)*A(i,k)/A(k,k);
0030        end
0031      end //i loop
0032    end //k loop
0033    //Backsubstitution loop
0034    x = zeros(n,1);
0035    x(n) = A(n,n+1)/A(n,n); 
0036    for i=n-1:-1:1
0037      x(i) = A(i,n+1);
0038      for j=i+1:n
0039        x(i) = x(i)-A(i,j)*x(j);
0040      end
0041      x(i) = x(i)/A(i,i);
0042    end
0043  endfunction

7 Appendix – Matlab code

7.1 Gauss-Jordan elimination
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%% linearGaussJordan.sci
%% 2014-06-23, Scott Hudson, for pedagogic purposes
%% Solves Ax=b for x using Gauss-Jordan elimination.
%% No pivoting is performed.
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
function x = linearGaussJordan(A,b)
  n = length(b);
  A = [A,b]; %%form augmented matrix
  for k=1:n %%A(k,k) is the pivot
    for j=n+1:-1:k %%normalize row k so A(k,k)=1
      A(k,j) = A(k,j)/A(k,k); 
    end
    for i=1:n %%eliminate a(i,k) for all i~=k
      if (i~=k) %%a Pivot does not eliminate itself
        for j=n+1:-1:k
          A(i,j) = A(i,j)-A(k,j)*A(i,k);
        end
      end
    end %%i loop
  end %%k loop
  x = A(:,n+1); %%last column of augmented matrix is now x
end

7.2 Matrix inverse using Gauss-Jordan elimination
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%% linearGaussJordanInverse.sci
%% 2014-06-23, Scott Hudson, for pedagogic purposes
%% Forms inverse of matrix A using Gauss-Jordan elimination.
%% No pivoting is performed.
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
function Ainv = linearGaussJordanInverse(A)
  n = size(A,1);
  A = [A,eye(size(A))]; %%form augmented matrix
  for k=1:n %%A(k,k) is the pivot
    for j=2*n:-1:k %%normalize row k so A(k,k)=1
      A(k,j) = A(k,j)/A(k,k); 
    end
    for i=1:n %%eliminate a(i,k) for all i~=k
      if (i~=k) %%a Pivot does not eliminate itself
        for j=2*n:-1:k
          A(i,j) = A(i,j)-A(k,j)*A(i,k);
        end
      end
    end %%i loop
  end %%k loop
  Ainv = A(:,n+1:2*n); %%last column of augmented matrix is now x
end

EE 221 Numerical Computing Scott Hudson 2015-08-18



Lecture 11: Linear systems of equations I 12/13

7.3 Gaussian elimination
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%% linearGaussian.sci
%% 2014-06-25, Scott Hudson, for pedagogic purposes
%% Solves Ax=b for x using Gaussian elimination and backsubstitution.
%% No pivoting is performed.
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
function x = linearGaussian(A,b)
  n = length(b);
  A = [A,b]; %%form augmented matrix
  %%Gaussian elimination loop
  for k=1:n-1 %%A(k,k) is the pivot
    for i=k+1:n %%eliminate a(i,k) for all i>k
      for j=n+1:-1:k
        A(i,j) = A(i,j)-A(k,j)*A(i,k)/A(k,k);
      end
    end %%i loop
  end %%k loop
  %%Backsubstitution loop
  x = zeros(n,1);
  x(n) = A(n,n+1)/A(n,n); 
  for i=n-1:-1:1
    x(i) = A(i,n+1);
    for j=i+1:n
      x(i) = x(i)-A(i,j)*x(j);
    end
    x(i) = x(i)/A(i,i);
  end
end

7.4 Gaussian elimination with partial pivoting
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%% linearGaussianPivot.sci
%% 2014-06-25, Scott Hudson, for pedagogic purposes
%% Solves Ax=b for x using Gaussian elimination and backsubstitution.
%% Partial pivoting is performed.
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
function x = linearGaussianPivot(A,b)
  n = length(b);
  A = [A,b]; %%form augmented matrix
  %%Gaussian elimination loop
  for k=1:n-1 %%A(k,k) is the pivot
    %%see if there is a larger pivot below this in the kth column
    Amax = abs(A(k,k));
    imax = k;
    for i=k+1:n
      if abs(A(i,k))>Amax
        Amax = abs(A(i,k));
        imax = i;
      end
    end
    if (imax~=k) %%we found a larger pivot, swap rows
      w = A(k,:); %%copy the kth row
      A(k,:) = A(imax,:); %%replace it with the imax row
      A(imax,:) = w; %%replace the imax row with the original kth row
    end
    %%pivoting complete
    for i=k+1:n %%eliminate a(i,k) for all i>k
      for j=n+1:-1:k
        A(i,j) = A(i,j)-A(k,j)*A(i,k)/A(k,k);
      end
    end %%i loop
  end %%k loop
  %%Backsubstitution loop
  x = zeros(n,1);
  x(n) = A(n,n+1)/A(n,n); 
  for i=n-1:-1:1
    x(i) = A(i,n+1);
    for j=i+1:n
      x(i) = x(i)-A(i,j)*x(j);
    end
    x(i) = x(i)/A(i,i);
  end
end

